HRP 262 SAS LAB SEVEN, May 23, 2012


Lab Seven: GEE; time independent vs. time-dependent predictors

Lab Objectives

After today’s lab you should be able to:

1. Analyze longitudinal data with GEE.

2. Interpret results from (1).
3. Analyze both continuous and binary predictors with GEE.

4. Understand the difference between time-independent and time-dependent predictors.

5. Interpret results with time-independent predictors

6. Understand the difference between “between-subjects” and “within subjects” effects. 

SAS PROCs                           SAS EG equivalent

PROC GENMOD

Analyze(Regression(Generalized Linear Regression

PROC GPLOT

Graph ((Line Plot)
 LAB EXERCISE STEPS:

Follow along with the computer in front…

1. For today’s class, download the lab 4-8 data at: www.stanford.edu/~kcobb/courses/hrp262. Make sure to re-download the data this time—I’ve added an additional variable. 
2. Open SAS EG; create a library pointing to the desktop.
3. Using code, turn the data into the long form, with both a continuous and categorical measure of time (time in months and dxa). Create both a repeated-measure outcome variable (bmc) and repeated-measure (=time-dependent) predictor (calcium). Do not fill in missing observations, since mixed models and GEE account for these.

data hrp262.runners;

set hrp262.runners;

id=_n_;

run;


data long;

set hrp262.runners;

dxa=1; time=0; bmc=bmc1; calc=calc1; injury=injury1; output;

dxa=2; time=(dxaday2-dxaday1)*12/365.25; bmc=bmc2; calc=calc2; injury=injury2; output;

dxa=3; time=(dxaday3-dxaday1)*12/365.25; bmc=bmc3; calc=calc3; injury=injury3; output;

label time='Months since baseline';

label bmc='BMC (g)';

label calc='dietary calcium, mg/day';

run;

4. Recall repeated measures ANOVA results and graphics from last time:

Predictor: treatment assignment:

Predictor: baseline calcium (divided into tertiles):

5. Now, look at the data graphically.  Last time we plotted BMC against time as categorical. Now see what happens if you plot BMC against continuous time.  
Go to the newly-created long dataset. Click on Graph > Line Plot. Like last time, we will have to modify the code once EG has created it, in order to get a better-appearing plot.

Choose Multiple line plots by group column.

Horizontal variable should be time and Vertical should be bmc. Group the plot by id. Next click on Edit so that we can filter our data by time.

Create a Task filter for time less than 30. We want to restrict the graph to 30 months of follow-up to avoid a lot of white space in the graph (women were supposed to finish the study in 2 years). This is equivalent to a ‘where time<30’ statement in SAS code.

Under Legend uncheck Show legend.

Like last time, the plot is not very appealing. We must modify the code directly to get the plot we want. To modify the code in EG, click on the Code tab and begin typing. When EG prompts you, click “Yes” to create a copy of the code. Looking over the code, you can see that EG has created new symbols for each observation. We will add a repeat statement in the code for the first symbol1, and then comment out the rest of the symbols so they are not used:

Click Run.

To do this entirely in SAS code:

/*With time as continuous*/
proc gplot data=long;

   symbol1 c=black i=join v=dot height=.5 repeat=78;

  plot bmc*time=id/nolegend vaxis=axis1;

  where time<30;

run; quit;

6. A better graph with continuous time is to plot each person’s percent change in BMC from baseline against time as follows:

data long2;

set hrp262.runners;

time=0; chgbmc=0; output;

time=(dxaday2-dxaday1)*12/365.25; chgbmc=(bmc2-bmc1)/bmc1; output;

time=(dxaday3-dxaday1)*12/365.25; chgbmc=(bmc3-bmc1)/bmc1; output;

label time='Months since baseline';

label chgbmc='% change in BMC since baseline';

run;

Graph(Scatter plot

2D scatter plot

Under Data, choose chgbmc as the vertical axis variable and time as the horizontal axis variable: 

Under plots,  change the plotting symbol to a red dot:

Under Interpolations, ask to overlay a smoothing line:

Then Click Run. 

We’re eventually going to look at three different ways to handle repeated measures data in SAS. (1) PROC GENMOD with a repeated statement (=Generalized Estimating Equations, or GEE); (2) PROC MIXED with a repeated statement (=mixed model for repeated measures); (3) PROC MIXED with a random statement (=mixed model with random effects). Today we will focus on the first option.
	Procedure
	Dependent variable
	Random effects?

	PROC GENMOD with repeated statement
	Any—binary, count, continuous, etc. 
	No

	PROC MIXED with repeated statement
	Continuous, normally distributed
	No

	PROC MIXED with a random statement
	Continuous, normally distributed
	Yes


7. But first, let’s examine the empirical variance/covariance and correlation matrices first (on the hrp262.runners dataset, broad form of the data):
Analyze > Multivariate > Correlations:

Under Data, select bmc1-3 as your Analysis variables.

Under Options, select Pearson correlation and the option to obtain Covariances.

Click Run.

	Variances and Covariances 
Covariance / Row Var Variance / Col Var Variance / DF

	 
	bmc1
	bmc2
	bmc3

	bmc1

 


	95687.5715

95687.5715

95687.5715

77


	89922.3892

91025.4984

90828.7027

72


	98149.6313

96567.1785

105636.5828

48



	bmc2

BMC2


	89922.3892

90828.7027

91025.4984

72


	90828.7027

90828.7027

90828.7027

72


	92403.6928

92241.3561

98760.0525

45



	bmc3

BMC3


	98149.6313

105636.5828

96567.1785

48


	92403.6928

98760.0525

92241.3561

45


	105636.5828

105636.5828

105636.5828

48




	Pearson Correlation Coefficients 
Prob > |r| under H0: Rho=0 
Number of Observations

	 
	bmc1
	bmc2
	bmc3

	bmc1

 


	1.00000

 

78


	0.98895

<.0001

73


	0.97178

<.0001

49



	bmc2

BMC2


	0.98895

<.0001

73


	1.00000

 

73


	0.96813

<.0001

46



	bmc3

BMC3


	0.97178

<.0001

49


	0.96813

<.0001

46


	1.00000

 

49




FYI, corresponding SAS code:

proc corr data=hrp262.runners cov;

var bmc1 bmc2 bmc3;

run;
8. Run a GEE model with treatment randomization as a predictor (time-independent predictor) using the long form of the data. 
From the Long form of the data(ANALYZE(REGRESSION(GENERALIZED LINEAR MODELS

Select BMC as the outcome variable and Treatr and Time as the quantitative predictor variables; AND choose id as a classification variable.

Include in the model main effects and a time*treatr interaction:

Specify that our outcome is normally distributed and we want the identity link function (this is the default):

Click Run. NOTE: there is no option to add a repeated statement in the point and click. We will have to do this using code. Go directly into the code and add the repeated statement:
PROC GENMOD DATA=WORK.SORTTempTableSorted



PLOTS(ONLY)=ALL
; 
CLASS id

MODEL bmc=
time treatr time*treatr



/



LINK=IDENTITY



DIST=NORMAL


;


repeated subject=id /type=exch corrw;

RUN; QUIT;
	Working Correlation Matrix

	 
	Col1
	Col2
	Col3

	Row1
	1.0000
	0.9715
	0.9715

	Row2
	0.9715
	1.0000
	0.9715

	Row3
	0.9715
	0.9715
	1.0000



	Analysis Of GEE Parameter Estimates

	Empirical Standard Error Estimates

	Parameter
	Estimate
	Standard Error
	95% Confidence Limits
	Z
	Pr > |Z|

	Intercept
	2174.489
	43.3302
	2089.563
	2259.414
	50.18
	<.0001

	time
	0.8620
	0.4930
	-0.1042
	1.8282
	1.75
	0.0804

	treatr
	21.3704
	71.8467
	-119.447
	162.1874
	0.30
	0.7661

	time*treatr
	0.1248
	0.7609
	-1.3666
	1.6161
	0.16
	0.8698




9. Now, run a model using baseline calcium (calc1) as the time-independent predictor rather than treatment group:
Analyze(Regression(Generalized Linear Models

Select bmc as the dependent variable; time and calc1 as the quantitative predictors; and id as the classification variables. 

In the Model screen, specify the model: time, calc1, and time*calc1. 

Under Model Options, ask for a normal distribution with an identity link.

Then Click Run. 

Then manually modify the code to add the repeated statement:

PROC GENMOD DATA=WORK.SORTTempTableSorted



PLOTS(ONLY)=ALL
;


CLASS id


;


MODEL bmc=
time calc1 time*calc1


/



LINK=IDENTITY



DIST=NORMAL


;


repeated subject=id/type=exch corrw;

RUN; QUIT;

FYI, corresponding SAS code:

proc genmod data=long;

class  id; 
model bmc=  calc1 time calc1*time; 

repeated subject = id / type=exch  corrw ;

run; quit;

	Working Correlation Matrix

	 
	Col1
	Col2
	Col3

	Row1
	1.0000
	0.9715
	0.9715

	Row2
	0.9715
	1.0000
	0.9715

	Row3
	0.9715
	0.9715
	1.0000



	GEE Fit Criteria

	QIC
	203.2460

	QICu
	202.0000


	Analysis Of GEE Parameter Estimates

	Empirical Standard Error Estimates

	Parameter
	Estimate
	Standard Error
	95% Confidence Limits
	Z
	Pr > |Z|

	Intercept
	2179.000
	76.3872
	2029.284
	2328.716
	28.53
	<.0001

	calc1
	0.0017
	0.0431
	-0.0828
	0.0862
	0.04
	0.9687

	time
	-0.8399
	0.6728
	-2.1586
	0.4789
	-1.25
	0.2119

	calc1*time
	0.0012
	0.0004
	0.0004
	0.0020
	3.03
	0.0024



9. Now, change the correlation structure to unstructured to see how that affects the model:

PROC GENMOD DATA=WORK.SORTTempTableSorted



PLOTS(ONLY)=ALL
;


CLASS id


;


MODEL bmc=
time calc1 time*calc1


/



LINK=IDENTITY



DIST=NORMAL


;


repeated subject=id/type=un corrw;
RUN; QUIT;
	Working Correlation Matrix

	 
	Col1
	Col2
	Col3

	Row1
	1.0000
	0.9839
	0.9959

	Row2
	0.9839
	1.0000
	0.9959

	Row3
	0.9959
	0.9959
	1.0000



	GEE Fit Criteria

	QIC
	208.8700

	QICu
	202.0000


	Analysis Of GEE Parameter Estimates

	Empirical Standard Error Estimates

	Parameter
	Estimate
	Standard Error
	95% Confidence Limits
	Z
	Pr > |Z|

	Intercept
	2046.697
	92.0754
	1866.232
	2227.161
	22.23
	<.0001

	time
	-1.0536
	0.9327
	-2.8817
	0.7746
	-1.13
	0.2587

	calc1
	0.1009
	0.0529
	-0.0027
	0.2045
	1.91
	0.0563

	time*calc1
	0.0014
	0.0005
	0.0003
	0.0024
	2.51
	0.0120


11. Now, let’s model calcium as a time-dependent predictor: using the calc variable. **Do not include an interaction term, since calc is time-dependent! Just use code to run this model:
proc genmod data=long;

class  id; 

model bmc= calc time ; 

repeated subject = id / type=exch corrw ;

run; quit;

	Analysis Of GEE Parameter Estimates

	Empirical Standard Error Estimates

	Parameter
	Estimate
	Standard Error
	95% Confidence Limits
	Z
	Pr > |Z|

	Intercept
	2185.019
	38.3723
	2109.810
	2260.227
	56.94
	<.0001

	calc
	-0.0053
	0.0136
	-0.0320
	0.0215
	-0.39
	0.6999

	time
	0.9131
	0.3956
	0.1376
	1.6885
	2.31
	0.0210


What’s happening is that your baseline BMC has nothing to do with your baseline calcium intake (since your skeletal size was determined as a teenager. There are huge between-subject differences in BMC, but they aren’t accounted for by current calcium intake. A simple scatterplot of bmc vs calc1 for each individual will reveal this large between-subject variation.Under the long dataset, go to Graph > Scatter Plot.

Select 2D Scatter Plot. Choose calc1 on the Horizontal and bmc1 on the Vertical axes.

Under Interpolations choose Linear Regression.

Click Run.

FYI, corresponding SAS cod:
axis2 label=(angle=90);

symbol1  c=red v=dot i=rl;

proc gplot data=hrp262.runners;

plot bmc1*calc1 /vaxis=axis1;

run;
This doesn’t rule out the possibility that there may be some within-subjects effects for calcium that are just overwhelmed by the lack of between-subjects effects.  In other words, a high calcium intake may still predict changes in BMC over time. But the within-subject variation is just tiny, tiny compared with the between-subject variation…

When you have a time-dependent predictor in GEE and mixed models, you cannot distinguish between between-subjects and within-subjects effects! What other strategies could you use to tease out between-subjects and within-subjects effects?
12. Now, let’s look at a binary outcome, injury. We must specify a distribution and link now! The analysis is akin to logistic regression. The descending option models “1” as the event and “0” as the non-event. Using code: 
proc genmod data=long descending;

class  id; 

model injury= calc time /DIST=binomial LINK=logit; 

repeated subject = id / type=exch corrw ;

run; quit;

	Working Correlation Matrix

	 
	Col1
	Col2
	Col3

	Row1
	1.0000
	-0.0583
	-0.0583

	Row2
	-0.0583
	1.0000
	-0.0583

	Row3
	-0.0583
	-0.0583
	1.0000


	Analysis Of GEE Parameter Estimates

	Empirical Standard Error Estimates

	Parameter
	Estimate
	Standard Error
	95% Confidence Limits
	Z
	Pr > |Z|

	Intercept
	0.2137
	0.3801
	-0.5313
	0.9587
	0.56
	0.5740

	calc
	-0.0004
	0.0002
	-0.0008
	0.0001
	-1.65
	0.0998

	time
	0.0056
	0.0133
	-0.0205
	0.0318
	0.42
	0.6728


13. The beta coefficients have an odds ratio interpretation, just like logistic regression. But we have to ask for the odds ratios explicitly. Add to code:

proc genmod data=long descending;

class  id; 

model injury= calc time /DIST=binomial LINK=logit; 

repeated subject = id / type=exch corrw ;

estimate "Beta for 100 mg calcium" calc 100 / exp;

run; quit;
	Contrast Estimate Results

	Label
	Mean Estimate
	Mean
	L'Beta Estimate
	Standard Error
	Alpha
	L'Beta
	Chi-Square
	Pr > ChiSq

	
	
	Confidence Limits
	
	
	
	Confidence Limits
	
	

	Odds ratio for 100 mg calcium
	0.4910
	0.4802
	0.5017
	-0.0362
	0.0220
	0.05
	-0.0792
	0.0069
	2.71
	0.0998

	Exp(Odds ratio for 100 mg calcium)
	 
	 
	 
	0.9645
	0.0212
	0.05
	0.9238
	1.0069
	 
	 


Every additional 100 mg of calcium in the diet decreases the odds of injury by 3.5%, 95% CI for the OR: .92 10 1.01.   
14. One additional note; when dealing with a low number of clusters, there is an alternate way to calculate the standard errors in GEE, called “model based standard errors.” To request these (recommended for the final exam, in which there are only 10 clusters), add the code:

proc genmod data=long descending;

class  id; 

model injury= calc time /DIST=binomial LINK=logit; 

repeated subject = id / type=exch corrw MODELSE;

estimate "Beta for 100 mg calcium" calc 100 / exp;

run; quit;

	Analysis Of GEE Parameter Estimates

	Model-Based Standard Error Estimates

	Parameter
	Estimate
	Standard Error
	95% Confidence Limits
	Z
	Pr > |Z|

	Intercept
	0.2137
	0.3639
	-0.4996
	0.9269
	0.59
	0.5571

	calc
	-0.0004
	0.0002
	-0.0008
	0.0000
	-1.72
	0.0848

	time
	0.0056
	0.0136
	-0.0210
	0.0322
	0.41
	0.6784

	Scale
	1.0000
	.
	.
	.
	.
	.


15. Also, use baseline calcium as a time-independent predictor in the model: 

proc genmod data=long descending;

class  id; 

model injury= calc1 time calc1*time /DIST=binomial LINK=logit; 

repeated subject = id / type=exch corrw;

run; quit;

	Analysis Of GEE Parameter Estimates

	Empirical Standard Error Estimates

	Parameter
	Estimate
	Standard Error
	95% Confidence Limits
	Z
	Pr > |Z|

	Intercept
	0.0807
	0.4400
	-0.7817
	0.9431
	0.18
	0.8544

	calc1
	-0.0002
	0.0003
	-0.0008
	0.0003
	-0.94
	0.3466

	time
	0.0709
	0.0314
	0.0094
	0.1325
	2.26
	0.0238

	calc1*time
	-0.0000
	0.0000
	-0.0001
	-0.0000
	-2.33
	0.0198


16. Finally, compare the use of conditional logistic regression to analyze these data:

proc logistic data=long;

class id;

model injury (event="1") = calc time;

strata id;

run;



The treatr term represents the baseline difference in bmc between the treatment and control groups. 





The treatr*time term represents the difference in the rate of change in bmc per month for the treatment vs. the control groups. 





Ask to see the working correlation matrix.











One of the QIC statistics is higher now, suggesting that exchangeable correlation matrix was a better choice. But results from the two models are similar








Gives a nice illustration of the small linear increase over time. 





Asks for unstructured correlation matrix. 























Three correlations are estimated now! 








Asks to see the estimated correlations.





Exchangeable = compound symmetry. 





Asks for exchangeable correlation matrix.








Similar results to rANOVA. But now we can quantify the rate of change in bmc:  almost 1 g/month in each group (.86g/month in the control group; .86+.12=.98g/month in the treatment group) 





The time term represents the rate of change in bmc per month (.86 grams/month) for the control group. (significant)








Conclusions: both the treatment and control groups have significant within-subject changes in bmc over time, but they are increasing at about the same rate = .86 g/month in the control group vs. .98 g/month in the treatment group. 








The intercept represents the average bmc at baseline for the control group. 








Tells SAS that we have repeated measures; and that the variable id identifies the subjects.








Note the assignment of permanent labels to variables within a datastep. This is a global assignment, so all subsequent plots and procedures will use these labels by default.


























Baseline calcium intake significantly predicts rate of change in bmc over time. If you have no calcium, your rate of change is projected to be -.73 grams/month. If you have 1000 mg of calcium, your rate of change is projected to be 1.084-.73 grams/month = .35 grams/month. Every 1000 mg calcium extra per day increases your rate of change in bmc by 1.084 grams/month. 





-Variance looks just slightly higher for bmc3.


-The three covariances/correlations look similar with the highest correlation between bmc1 and bmc2.


-Looks like compound symmetry is relatively well satisfied; we could also try the unstructured. 








By rANOVA and rMANOVA,


Time (DXA) is significant


Calcium is not significant


Time*Calcium is significant





By rANOVA and rMANOVA,


Time (DXA) is significant


Treatr is not significant


Time*Treatr is not significant








The correlations for injury are actually fairly low. 








We need to assign a unique ID number to each participant for the long form (otherwise we will be unable to tell which observations belong to which subject)!





The exchangeable correlation matrix. 





Calcium is a time-dependent predictor.





QIC statistic can help you choose between models; select the model with the lowest QIC.





I’m restricting the graph to 30 months of follow-up, to avoid a lot of white space in the graph (women were supposed to finish the study in 2 years).





Continuous time is great for modeling, but bad for graphics!!

